



### program:



```
program ::= ( statement | NEWLINE )* EOF
```

no references

### statement:




```
statement
    ::= declaration_stmtnt
       | import_stmt
       | print_stmtnt
       | input_stmtnt
       | return_stmtnt
       | if_stmtnt
       | while_stmtnt
       | do_stmtnt
       | for_stmtnt
       | break_stmtnt
       | continue_stmtnt
       | pass_stmtnt
       | expression_stmtnt
```

referenced by:

- [block](#)
- [program](#)

### declaration\_stmt:




```
declaration_stmt  
  ::= variable_declaration  
     | function_declaration
```

referenced by:

- [statement](#)

### variable\_declaration:




```
variable_declaration  
  ::= var_type identifier ( '=' assignment_expr )? ( ',' identifier ( '=' assignment_expr )? )* NEWLINE
```

referenced by:

- [declaration\\_stmt](#)

### var\_type:




```
var_type ::= 'char'  
         | 'int'  
         | 'float'  
         | 'str'  
         | 'list'
```

referenced by:

- variable declaration

### function\_declaration:




```
function_declaration
    ::= 'def' identifier '(' ( identifier ( ',' identifier )* )? ')' block
```

referenced by:

- declaration\_stmtnt

### identifier:




```
identifier
    ::= alphabetic ( alphabetic | digit | '_' )*
```

referenced by:

- function\_call
- function\_declaration
- input\_stmtnt
- variable\_declaration

### block:




```
block    ::= NEWLINE INDENT statement+ DEDENT
```

referenced by:

- do\_stmtnt
- function\_declaration
- if\_stmtnt
- while\_stmtnt

### import\_stmt:




```
import_stmt
    ::= 'import' assignment_expr ( ',' assignment_expr )* NEWLINE
```

referenced by:

- [statement](#)

### print\_stmt:




```
print_stmt
    ::= 'print' assignment_expr ( ',' assignment_expr )* NEWLINE
```

referenced by:

- [statement](#)

### input\_stmt:




```
input_stmt
    ::= 'input' string? identifier ( ',' string? identifier )* NEWLINE
```

referenced by:

- [statement](#)

### return\_stmt:




```
return_stmt
    ::= 'return' expression? NEWLINE
```

referenced by:

- [statement](#)

### if\_stmt:




```
if_stmt ::= 'if' expression block ( 'else' block )?
```

referenced by:

- [statement](#)

### while\_stmt:



```
while_stmt  
    ::= 'while' expression block
```

referenced by:

- [statement](#)

### do\_stmt:



```
do_stmt ::= 'do' block 'while' expression NEWLINE
```

referenced by:

- [statement](#)

### break\_stmt:



```
break_stmt  
    ::= 'break' NEWLINE
```

referenced by:

- [statement](#)

### continue\_stmt:




```
continue_stmt  
    ::= 'continue' NEWLINE
```

referenced by:

- [statement](#)

### pass\_stmtnt:




```
pass_stmtnt
  ::= 'pass' NEWLINE
```

referenced by:

- [statement](#)

### for\_stmtnt:



```
for_stmtnt
  ::= 'for' IDENTIFIER 'in' sequence
```

referenced by:

- [statement](#)

### expression\_stmtnt:




```
expression_stmtnt
  ::= expression NEWLINE
```

referenced by:

- [statement](#)

### expression:




```
expression
  ::= assignment_expr ( ',' expression )*
```

referenced by:

- [do\\_stmtnt](#)
- [expression](#)
- [expression\\_stmtnt](#)
- [if\\_stmtnt](#)
- [primary\\_expr](#)
- [return\\_stmtnt](#)
- [while\\_stmtnt](#)

## assignment\_expr:




```
assignment_expr
    ::= logical_or_expr ( ( '=' | '+=' | '-=' | '*=' | '\\=' | '%=' ) assignment_expr )*
```

referenced by:

- [assignment\\_expr](#)
- [expression](#)
- [function\\_call](#)
- [import\\_stmt](#)
- [list\\_const](#)
- [print\\_stmt](#)
- [variable\\_declaration](#)

## logical\_or\_expr:




```
logical_or_expr
    ::= logical_and_expr ( 'or' logical_or_expr )*
```

referenced by:

- [assignment\\_expr](#)
- [index](#)
- [list\\_append](#)
- [list\\_insert](#)
- [logical\\_or\\_expr](#)
- [slice](#)

## logical\_and\_expr:




```
logical_and_expr
    ::= equality_expr ( 'and' logical_and_expr )*
```

referenced by:

- logical and expr
- logical or expr

### equality\_expr:




```
equality_expr
    ::= relational_expr ( ( '=' | '!=' | '<>' ) equality_expr )*
```

referenced by:

- equality\_expr
- logical and expr

### relational\_expr:




```
relational_expr
    ::= additive_expr ( ( '<' | '>' | '<=' | '>=' ) relational_expr )*
```

referenced by:

- equality\_expr
- relational\_expr

### additive\_expr:




```
additive_expr
    ::= mult_expr ( ( '+' | '-' ) additive_expr )*
```

referenced by:



- additive\_expr
- relational\_expr

### mult\_expr:




```
mult_expr
    ::= unary_expr ( ( '*' | '/' | '%' ) mult_expr )*
```

referenced by:

- additive\_expr
- mult\_expr

### unary\_expr:




```
unary_expr
    ::= ( '+' | '-' | '!' )? primary_expr
```

referenced by:

- mult\_expr

### primary\_expr:




```
primary_expr
    ::= function_call
       | variable
       | constant
       | '(' expression ')'
```

referenced by:

- [unary\\_expr](#)

### function\_call:




```
function_call
    ::= identifier '(' ( assignment_expr ( ',' assignment_expr )* )? ')'
```

referenced by:

- [primary\\_expr](#)

### variable:




```
variable ::= ( numeric_variable | sequence_variable ) ( '.' method )?
```

referenced by:

- [primary\\_expr](#)

### numeric\_variable:




```
numeric_variable
    ::= 'identifier of variable of type char'
    | 'identifier of variable of type int'
    | 'identifier of variable of type float'
```

referenced by:

- [variable](#)

### sequence\_variable:




```
sequence_variable
    ::= ( 'identifier of variable of type str' | 'identifier of variable of type list' ) subscript?
```

referenced by:

- [variable](#)

### sequence:




```
sequence ::= ( 'identifier of variable of type str' | 'identifier of variable of type list' ) ( '[' slice ']' )?
```

referenced by:

- [for\\_stmt](#)

### method:



```
method    ::= 'type'
            | list_insert
            | list_append
            | list_remove
            | 'len'
```

referenced by:

- [variable](#)

### list\_insert:




```
list_insert
    ::= 'insert' '(' index ',' logical_or_expr ')'
```

referenced by:

- [method](#)

### list\_append:




```
list_append
    ::= 'append' '(' logical_or_expr ')'
```

referenced by:

- [method](#)

### list\_remove:




```
list_remove
    ::= 'remove' '(' index ')'
```

referenced by:

- [method](#)

### subscript:




```
subscript
    ::= '[' ( index | slice ) ']'
```

referenced by:

- [sequence\\_variable](#)

### index:




```
index    ::= logical_or_expr
```

referenced by:

- [list\\_insert](#)
- [list\\_remove](#)
- [subscript](#)

### slice:




```
slice ::= logical_or_expr? ':' logical_or_expr?
```

referenced by:

- [sequence](#)
- [subscript](#)

### constant:




```
constant ::= numeric_const  
          | character_const  
          | string_const  
          | list_const
```

referenced by:

- [primary\\_expr](#)

### numeric\_const:




```
numeric_const  
  ::= number
```

referenced by:

- [constant](#)

### character\_const:




```
character_const  
  ::= "'" character "'"
```

referenced by:

- [constant](#)

### string\_const:




```
string_const
    ::= string
```

referenced by:

- [constant](#)

### list\_const:




```
list_const
    ::= '[' ( assignment_expr ( ',' assignment_expr )* )? ']
```

referenced by:

- [constant](#)

### string:




```
string    ::= ''' character* '''
```

referenced by:

- [input\\_stmt](#)
- [string\\_const](#)

### character:




```
character
    ::= 'any character except single quote'
    | escape
```

referenced by:

- [character\\_const](#)
- [string](#)

### alphanumeric:




```
alphabetic  
  ::= [a-zA-Z]
```

referenced by:

- identifier

### escape:




```
escape  ::= '\\' [bnfrtv\\'\"0]
```

referenced by:

- character

### number:




```
number  ::= integer  
          | float
```

referenced by:

- numeric const

### integer:




```
integer ::= ( '+' | '-' )? digit+
```

referenced by:

- [number](#)

### float:




```
float ::= ( '+' | '-' )? digit+ '.' digit* ( ( 'e' | 'E' ) ( '+' | '-' )? digit+ )?
```

referenced by:

- [number](#)

### digit:



```
digit ::= [0-9]
```

referenced by:

- [float](#)
- [identifier](#)
- [integer](#)